NAG Library Routine Document

E04YAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

E04YAF checks that a user-supplied subroutine for evaluating a vector of functions and the matrix of their first derivatives produces derivative values which are consistent with the function values calculated.

2 Specification

```
SUBROUTINE E04YAF(M, N, LSQFUN, X, FVEC, FJAC, LDFJAC, IW, LIW, W, LW,
1                 IFAIL)
INTEGER           M, N, LDFJAC, IW(LIW), LIW, LW, IFAIL
double precision  X(N), FVEC(M), FJAC(LDFJAC,N), W(LW)
EXTERNAL          LSQFUN
```

3 Description

Routines for minimizing a sum of squares of m nonlinear functions (or 'residuals'), $f_i(x_1, x_2, \ldots, x_n)$, for $i = 1, 2, \ldots, m$ and $m \geq n$, may require you to supply a subroutine to evaluate the f_i and their first derivatives. E04YAF checks the derivatives calculated by such user-supplied subroutines, e.g., routines of the form required for E04GBF, E04GDF and E04HEF. As well as the routine to be checked (LSQFUN), you must supply a point $x = (x_1, x_2, \ldots, x_n)^{\mathrm{T}}$ at which the check will be made. E04YAF is essentially identical to CHKLSJ in the NPL Algorithms Library.

E04YAF first calls LSQFUN to evaluate the $f_i(x)$ and their first derivatives, and uses these to calculate the sum of squares $F(x) = \sum_{i=1}^{m} [f_i(x)]^2$, and its first derivatives $g_j = \left.\dfrac{\partial F}{\partial x_j}\right|_x$, for $j = 1, 2, \ldots, n$. The components of g along two orthogonal directions (defined by unit vectors p_1 and p_2, say) are then calculated; these will be $g^{\mathrm{T}}p_1$ and $g^{\mathrm{T}}p_2$ respectively. The same components are also estimated by finite differences, giving quantities

$$v_k = \frac{F(x + hp_k) - F(x)}{h}, \qquad k = 1, 2$$

where h is a small positive scalar. If the relative difference between v_1 and $g^{\mathrm{T}}p_1$ or between v_2 and $g^{\mathrm{T}}p_2$ is judged too large, an error indicator is set.

4 References

None.

5 Parameters

1: M – INTEGER *Input*

2: N – INTEGER *Input*

On entry: the number m of residuals, $f_i(x)$, and the number n of variables, x_j.

Constraint: $1 \leq \mathrm{N} \leq \mathrm{M}$.

3: LSQFUN – SUBROUTINE, supplied by the user. *External Procedure*

LSQFUN must calculate the vector of values $f_i(x)$ and their first derivatives $\dfrac{\partial f_i}{\partial x_j}$ at any point x. (The minimization routines mentioned in Section 3 give you the option of resetting a parameter to terminate immediately. E04YAF will also terminate immediately, without finishing the checking process, if the parameter in question is reset.)

The specification of LSQFUN is:

```
SUBROUTINE LSQFUN(IFLAG, M, N, XC, FVEC, FJAC, LDFJAC, IW, LIW, W,
1                 LW)
INTEGER           IFLAG, M, N, LDFJAC, IW(LIW), LIW, LW
double precision  XC(N), FVEC(M), FJAC(LDFJAC,N), W(LW)
```

1: IFLAG – INTEGER *Input/Output*

On entry: to LSQFUN, IFLAG will be set to 2.

On exit: if you reset IFLAG to some negative number in LSQFUN and return control to E04YAF, the routine will terminate immediately with IFAIL set to your setting of IFLAG.

2: M – INTEGER *Input*
3: N – INTEGER *Input*

On entry: the numbers m and n of residuals and variables, respectively.

4: XC(N) – ***double precision*** array *Input*

On entry: x, the point at which the values of the f_i and the $\dfrac{\partial f_i}{\partial x_j}$ are required.

5: FVEC(M) – ***double precision*** array *Output*

On exit: unless IFLAG is reset to a negative number, FVEC(i) must contain the value of f_i at the point x, for $i = 1, 2, \ldots, m$.

6: FJAC(LDFJAC,N) – ***double precision*** array *Output*

On exit: unless IFLAG is reset to a negative number, FJAC(i, j) must contain the value of $\dfrac{\partial f_i}{\partial x_j}$ at the point x, for $i = 1, 2, \ldots, m$ and $j = 1, 2, \ldots, n$.

7: LDFJAC – INTEGER *Input*

On entry: the first dimension of the array FJAC as declared in the (sub)program from which E04YAF is called.

8: IW(LIW) – INTEGER array *Workspace*
9: LIW – INTEGER *Input*
10: W(LW) – ***double precision*** array *Workspace*
11: LW – INTEGER *Input*

These parameters are present so that LSQFUN will be of the form required by the minimization routines mentioned in Section 3. LSQFUN is called with the same parameters IW, LIW, W, LW as in the call to E04YAF. If the recommendation in the minimization routine document is followed, you will have no reason to examine or change the elements of IW or W. In any case, LSQFUN **must not change** the first $3 \times \mathrm{N} + \mathrm{M} + \mathrm{M} \times \mathrm{N}$ elements of W.

LSQFUN must be declared as EXTERNAL in the (sub)program from which E04YAF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

4: X(N) – ***double precision*** array *Input*

On entry: X(j), for $j = 1, 2, \ldots, n$, must be set to the co-ordinates of a suitable point at which to check the derivatives calculated by LSQFUN. 'Obvious' settings, such as 0 or 1, should not be used since, at such particular points, incorrect terms may take correct values (particularly zero), so that errors can go undetected. For a similar reason, it is preferable that no two elements of X should have the same value.

5: FVEC(M) – ***double precision*** array *Output*

On exit: unless you set IFLAG negative in the first call of LSQFUN, FVEC(i) contains the value of f_i at the point supplied by you in X, for $i = 1, 2, \ldots, m$.

6: FJAC(LDFJAC,N) – ***double precision*** array *Output*

On exit: unless you set IFLAG negative in the first call of LSQFUN, FJAC(i, j) contains the value of the first derivative $\dfrac{\partial f_i}{\partial x_j}$ at the point given in X, as calculated by LSQFUN, for $i = 1, 2, \ldots, m$ and $j = 1, 2, \ldots, n$.

7: LDFJAC – INTEGER *Input*

On entry: the first dimension of the array FJAC as declared in the (sub)program from which E04YAF is called.

Constraint: LDFJAC $\geq$ M.

8: IW(LIW) – INTEGER array *Communication Array*

This array appears in the parameter list purely so that, if E04YAF is called by another library routine, the library routine can pass quantities to LSQFUN via IW. IW is not examined or changed by E04YAF. In general you must provide an array IW, but are advised not to use it.

9: LIW – INTEGER *Input*

On entry: the dimension of the array IW as declared in the (sub)program from which E04YAF is called.

Constraint: LIW ≥ 1.

10: W(LW) – ***double precision*** array *Communication Array*

11: LW – INTEGER *Input*

On entry: the dimension of the array W as declared in the (sub)program from which E04YAF is called.

Constraint: LW $\geq 3 \times$ N + M + M $\times$ N.

12: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: E04YAF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL < 0

A negative value of IFAIL indicates an exit from E04YAF because you have set IFLAG negative in LSQFUN. The setting of IFAIL will be the same as your setting of IFLAG. The check on LSQFUN will not have been completed.

IFAIL $= 1$

On entry, M $<$ N,
or N $<$ 1,
or LDFJAC $<$ M,
or LIW $<$ 1,
or LW $<$ 3 $\times$ N + M + M $\times$ N.

IFAIL $= 2$

You should check carefully the derivation and programming of expressions for the $\dfrac{\partial f_i}{\partial x_j}$, because it is very unlikely that LSQFUN is calculating them correctly.

7 Accuracy

IFAIL is set to 2 if

$$\left(v_k - g^{\mathrm{T}} p_k\right)^2 \geq h \times \left(\left(g^{\mathrm{T}} p_k\right)^2 + 1\right)$$

for $k = 1$ or 2. (See Section 3 for definitions of the quantities involved.) The scalar h is set equal to $\sqrt{\epsilon}$, where ϵ is the ***machine precision*** as given by X02AJF.

8 Further Comments

E04YAF calls LSQFUN three times.

Before using E04YAF to check the calculation of the first derivatives, you should be confident that LSQFUN is calculating the residuals correctly.

E04YAF only checks the derivatives calculated by a user-supplied routine when IFLAG $= 2$. So, if LSQFUN is intended for use in conjunction with a minimization routine which may set IFLAG to 1, you must check that, for given settings of the XC(j), LSQFUN produces the same values for the $\dfrac{\partial f_i}{\partial x_j}$ when IFLAG is set to 1 as when IFLAG is set to 2.

9 Example

Suppose that it is intended to use E04GBF or E04GDF to find least-squares estimates of x_1, x_2 and x_3 in the model

$$y = x_1 + \frac{t_1}{x_2 t_2 + x_3 t_3}$$

using the 15 sets of data given in the following table.

y	t_1	t_2	t_3
0.14	1.0	15.0	1.0
0.18	2.0	14.0	2.0
0.22	3.0	13.0	3.0
0.25	4.0	12.0	4.0
0.29	5.0	11.0	5.0
0.32	6.0	10.0	6.0
0.35	7.0	9.0	7.0
0.39	8.0	8.0	8.0
0.37	9.0	7.0	7.0
0.58	10.0	6.0	6.0
0.73	11.0	5.0	5.0
0.96	12.0	4.0	4.0
1.34	13.0	3.0	3.0
2.10	14.0	2.0	2.0
4.39	15.0	1.0	1.0

The following program could be used to check the first derivatives calculated by LSQFUN. (The tests of whether IFLAG = 0 or 1 in LSQFUN are present ready for when LSQFUN is called by E04GBF or E04GDF. E04YAF will always call LSQFUN with IFLAG set to 2.)

9.1 Program Text

```
*     E04YAF Example Program Text
*     Mark 14 Revised. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          MDEC, NDEC, LDFJAC, LIW, LW
      PARAMETER        (MDEC=15,NDEC=3,LDFJAC=MDEC,LIW=1,
     +                 LW=3*NDEC+MDEC+MDEC*NDEC)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Arrays in Common ..
      DOUBLE PRECISION T(MDEC,NDEC), Y(MDEC)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, J, M, N
*     .. Local Arrays ..
      DOUBLE PRECISION FJAC(LDFJAC,NDEC), FVEC(MDEC), W(LW), X(NDEC)
      INTEGER          IW(LIW)
*     .. External Subroutines ..
      EXTERNAL         E04YAF, LSQFUN
*     .. Common blocks ..
      COMMON           Y, T
*     .. Executable Statements ..
      WRITE (NOUT,*) 'E04YAF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      N = NDEC
      M = MDEC
*     Observations of TJ (J = 1, 2, 3) are held in T(I, J)
*     (I = 1, 2, . . . , 15)
      DO 20 I = 1, M
         READ (NIN,*) Y(I), (T(I,J),J=1,N)
   20 CONTINUE
*     Set up an arbitrary point at which to check the 1st
*     derivatives
      X(1) = 0.19D0
      X(2) = -1.34D0
      X(3) = 0.88D0
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'The test point is'
      WRITE (NOUT,99999) (X(J),J=1,N)
      IFAIL = 1
*
      CALL E04YAF(M,N,LSQFUN,X,FVEC,FJAC,LDFJAC,IW,LIW,W,LW,IFAIL)
```

```
*
      WRITE (NOUT,*)
      IF (IFAIL.LT.0) THEN
         WRITE (NOUT,99998) IFAIL
      ELSE IF (IFAIL.EQ.1) THEN
         WRITE (NOUT,*) 'A parameter is outside its expected range'
      ELSE
         IF (IFAIL.EQ.0) THEN
            WRITE (NOUT,*)
     +        '1st derivatives are consistent with residual values'
         ELSE IF (IFAIL.EQ.2) THEN
            WRITE (NOUT,*)
     +        'Probable error in calculation of 1st derivatives'
         END IF
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'At the test point, LSQFUN gives'
         WRITE (NOUT,*)
         WRITE (NOUT,*)
     +     '      Residuals                    1st derivatives'
         WRITE (NOUT,99997) (FVEC(I),(FJAC(I,J),J=1,N),I=1,M)
      END IF
*
99999 FORMAT (1X,4F10.5)
99998 FORMAT (1X,' ** E04YAF returned with IFAIL = ',I5)
99997 FORMAT (1X,1P,4E15.3)
      END
*
      SUBROUTINE LSQFUN(IFLAG,M,N,XC,FVEC,FJAC,LDFJAC,IW,LIW,W,LW)
*     Routine to evaluate the residuals and their 1st derivatives
*     .. Parameters ..
      INTEGER          MDEC, NDEC
      PARAMETER        (MDEC=15,NDEC=3)
*     .. Scalar Arguments ..
      INTEGER          IFLAG, LDFJAC, LIW, LW, M, N
*     .. Array Arguments ..
      DOUBLE PRECISION FJAC(LDFJAC,N), FVEC(M), W(LW), XC(N)
      INTEGER          IW(LIW)
*     .. Arrays in Common ..
      DOUBLE PRECISION T(MDEC,NDEC), Y(MDEC)
*     .. Local Scalars ..
      DOUBLE PRECISION DENOM, DUMMY
      INTEGER          I
*     .. Common blocks ..
      COMMON           Y, T
*     .. Executable Statements ..
      DO 20 I = 1, M
         DENOM = XC(2)*T(I,2) + XC(3)*T(I,3)
         IF (IFLAG.NE.1) FVEC(I) = XC(1) + T(I,1)/DENOM - Y(I)
         IF (IFLAG.NE.0) THEN
            FJAC(I,1) = 1.0D0
            DUMMY = -1.0D0/(DENOM*DENOM)
            FJAC(I,2) = T(I,1)*T(I,2)*DUMMY
            FJAC(I,3) = T(I,1)*T(I,3)*DUMMY
         END IF
   20 CONTINUE
      RETURN
      END
```

9.2 Program Data

```
E04YAF Example Program Data
 0.14  1.0 15.0  1.0
 0.18  2.0 14.0  2.0
 0.22  3.0 13.0  3.0
 0.25  4.0 12.0  4.0
 0.29  5.0 11.0  5.0
 0.32  6.0 10.0  6.0
 0.35  7.0  9.0  7.0
 0.39  8.0  8.0  8.0
 0.37  9.0  7.0  7.0
```

```
0.58 10.0  6.0  6.0
0.73 11.0  5.0  5.0
0.96 12.0  4.0  4.0
1.34 13.0  3.0  3.0
2.10 14.0  2.0  2.0
4.39 15.0  1.0  1.0
```

9.3 Program Results

```
E04YAF Example Program Results

The test point is
   0.19000  -1.34000   0.88000

1st derivatives are consistent with residual values

At the test point, LSQFUN gives

      Residuals                    1st derivatives
     -2.029E-03      1.000E+00     -4.061E-02     -2.707E-03
     -1.076E-01      1.000E+00     -9.689E-02     -1.384E-02
     -2.330E-01      1.000E+00     -1.785E-01     -4.120E-02
     -3.785E-01      1.000E+00     -3.043E-01     -1.014E-01
     -5.836E-01      1.000E+00     -5.144E-01     -2.338E-01
     -8.689E-01      1.000E+00     -9.100E-01     -5.460E-01
     -1.346E+00      1.000E+00     -1.810E+00     -1.408E+00
     -2.374E+00      1.000E+00     -4.726E+00     -4.726E+00
     -2.975E+00      1.000E+00     -6.076E+00     -6.076E+00
     -4.013E+00      1.000E+00     -7.876E+00     -7.876E+00
     -5.323E+00      1.000E+00     -1.040E+01     -1.040E+01
     -7.292E+00      1.000E+00     -1.418E+01     -1.418E+01
     -1.057E+01      1.000E+00     -2.048E+01     -2.048E+01
     -1.713E+01      1.000E+00     -3.308E+01     -3.308E+01
     -3.681E+01      1.000E+00     -7.089E+01     -7.089E+01
```